United States
Naval Postgraduate School

THESIS

Response of a Water Column
to
Internal Waves of Known Frequencies

by

William John Lounsbery

Thesis Advisor: J.A. Galt

September 1971

United States
Naval Postgraduate School

THESIS

Response of a Water Column
to
Internal Waves of Known Frequencies

by

William John Lounsbery

Thesis Advisor: J.A. Galt

September 1971

Response of a Water Column
to
Internal Waves of Known Frequencies

by

William John Lounsbery
Lieutenant Commander, United States Navy
B.S., Auburn University, 1963

Submitted in partial fulfillment of the
requirements for the degree of

MASTER OF SCIENCE IN OCEANOGRAPHY

from the

NAVAL POSTGRADUATE SCHOOL
September 1971

Thesis

ABSTRACT

A program is developed to ascertain the response of a frictionless water column to internal waves. Using density strata as input, the program selects ten frequencies at equal intervals within a spectrum of internal waves bounded above by the maximum Vaisala frequency and below by the inertial frequency. Ray paths within the column are plotted for these frequencies. The first ten normal modes for each frequency are computed. The first three modes are plotted.

TABLE OF CONTENTS

LIST OF TABLES

LIST OF FIGURES

ACKNOWLEDGEMENT

I sincerely thank Dr. Jerry Galt, friend as well as advisor, who not only presented the initial problem, packaged neatly, but who also spent many extra hours helping me unwrap it. To my wife, Janice, I give special thanks for aiding me in drafting the thesis and, particularly, for weathering the storm of neglect while I concentrated on the problem.

I. INTRODUCTION

Internal waves exist in all oceans, most bays and lakes; they vary widely in amplitude, period, and depth (La Fond, 1962). Recently, advances have been made in interpreting the major features of the oceanic internal wave field (Wunsch and Dahlen, 1970).

Internal waves can be observed as density perturbations through a stable, stratified water column. Since temperature correlates with density in the majority of the upper ocean areas, internal waves often appear as oscillations about the region of the thermocline, exhibiting greatest amplitude and phase speed where the thermocline is the most intense. The oscillation initiated within the water column is described in terms of the Vaisala or stability frequency. This frequency may be derived as follows (Zalkan, 1966).

Consider the force acting on a water parcel raised vertically from its equilibrium position:

$$F = g[\rho - (\rho + \frac{\partial \rho}{\partial Z} \Delta Z)]dV - \rho \frac{g^2}{c^2} \Delta Z \, dV \qquad (1)$$

where: g = gravitational acceleration,

ρ = density,

ΔZ = vertical displacement,

dV = minute volume of the parcel, and

c = sound velocity. The first term on the right hand side is the restoring force due to buoyancy. The second term is the contributing force of compressibility. The equation

for the induced vertical motion is:

$$\rho dV \frac{dZ^2}{d^2Z} = -g \frac{\partial \rho}{\partial Z} ZdV - \rho \frac{g^2}{c^2} ZdV \qquad (2)$$

The harmonic solution to equation (2) is the Vaisala frequency defined as

$$N(Z) = \sqrt{\frac{-g}{\rho} \frac{\partial \rho}{\partial Z} - \frac{g^2}{c^2}} \qquad (3)$$

For oceanographic applications the water is effectively incompressible and the second term under the radical is negligible. Thus,

$$N(Z) = \sqrt{\frac{-g}{\rho} \frac{\partial \rho}{\partial Z}} \qquad (4)$$

The Vaisala frequency is greatest in magnitude where the thermocline is most intense (see Figure 1).

The vertical motion initiated by the internal wave passing through the water column is governed by the differential equation (Fjeldstad, 1933):

$$\frac{\partial}{\partial Z}\left(\rho \frac{dW}{dZ}\right) + \frac{k^2}{\sigma^2 - f^2}(N^2 - \sigma^2)W = 0 \qquad (5)$$

where: W = vertical water velocity,

k = internal wave number,

σ = internal wave frequency, and

f = inertial frequency. Equation (5) is a hyperbolic wave equation whose characteristics describe the direction of propagation of internal wave energy. With the appropriate boundary conditions that no vertical flow exists at the surface or at the bottom, the internal wave equation satisfies the criteria for a Sturmian system. The imposed boundary



FIGURE 1. Variation of Vaisala Frequency with Temperature Profile

conditions are satisfied by the simultaneous propagation of two waves having the same horizontal wave number, one with its vertical component of phase velocity up, the other with its vertical component down. The stipulation that these waves cancel at the boundaries leads to the structure of normal modes. These modes constitute a discrete set of wave numbers that are the eigenvalues of the Sturmian system.

This thesis, using as a basis Fjeldstad's classic paper (Fjeldstad, 1933), developed a numerical program that assessed the response by a frictionless, incompressible water column to internal waves in a frequency band bounded above by the maximum Vaisala frequency and below by the inertial frequency. Station density strata was used as input. Frequencies were selected within the band and their ray paths were drawn. Also, the governing differential equation was numerically integrated and the resulting endpoints were refined to determine the first ten normal modes. The first three modes were plotted.

The program was first tested using density strata with known output. The program then treated data taken from ocean stations off the coast of Monterey, California in mid-August, 1970.

II. THEORY

A. DEVELOPMENT OF THE INTERNAL WAVE EQUATION

For a frictionless, incompressible water column the governing equations (following Feldstad, 1933) are:

Momentum

$$\frac{\partial U}{\partial t} - fV + \frac{1}{\rho}\frac{\partial P}{\partial X} = 0 \tag{6}$$

$$\frac{\partial V}{\partial t} + fU + \frac{1}{\rho}\frac{\partial P}{\partial Y} = 0 \tag{7}$$

$$\frac{\partial W}{\partial t} + g + \frac{1}{\rho}\frac{\partial P}{\partial Z} = 0 \tag{8}$$

Continuity

$$\frac{\partial U}{\partial X} + \frac{\partial V}{\partial Y} + \frac{\partial W}{\partial Z} = 0 \tag{9}$$

Density invariant

$$\frac{\partial \rho}{\partial t} + U\frac{\partial \rho}{\partial x} + V\frac{\partial \rho}{\partial Y} + W\frac{\partial \rho}{\partial Z} = 0 \tag{10}$$

The X-axis is directed in the direction of wave travel, the Y-axis is in the horizontal transverse plane of travel, and the Z-axis is vertical, positive upward. U, V, and W are the velocity components of the water in the X, Y, and Z axes, respectively. f is the coriolis parameter. P is the pressure. The depth is assumed to be uniform. $Z = 0$ is the bottom; $Z = h$ is the undisturbed upper surface. For a steady state condition,

$$\frac{dU}{dt} = \frac{dV}{dt} = \frac{dW}{dt} = 0. \tag{11}$$

Thus,

$$dP = -\rho g dZ. \tag{12}$$

P can be represented as

$$P = P_0 + g\int_z^h \rho \, dz, \tag{13}$$

where P_0 is the undisturbed pressure at the surface. Let

$$\rho = \rho_0(Z) + \rho_1(X,Y,Z,t), \tag{14}$$

where ρ_0 is the undisturbed density and ρ_1 the perturbation. Therefore, pressure for the perturbation is:

$$P = P_0 + g\int_z^h \rho_0 \, dZ + P_1 \tag{15}$$

where

$$P_1 = g\int_h^z \rho_1 \, dZ \tag{16}$$

Envoking a general assumption that velocities and perturbations of density and pressure are small enough that their products and squares may be neglected, equation (10) may be written:

$$\frac{1}{\rho_0}\frac{\partial P_1}{\partial t} + \frac{1}{\rho_0}\frac{\partial \rho_0}{\partial Z} W = 0 \tag{17}$$

If, for the internal wave form, harmonic oscillations such as the following are considered:

$$u, \rho_1, P_1 \approx \cos\ (\sigma t - kx)\ \exp\ (ay)$$
$$V, W \approx \sin\ (\sigma t - kx)\ \exp\ (ay)$$

where the constant a controls the transverse envelope and allows for the possibility of edge waves. The sign of a is such that the wave amplitude decays away from the boundary. The governing equations now become:

$$\sigma U - fV + k\frac{P_1}{\rho_0} = 0 \tag{18}$$

$$\sigma V + fU + a\frac{P_1}{\rho_0} = 0 \tag{19}$$

$$\sigma W + g + \frac{1}{\rho_0}\frac{\partial P_1}{\partial Z} = 0 \qquad (20)$$

$$kU + aV + \frac{dW}{dZ} = 0 \qquad (21)$$

Solving equations (18) and (19) for U and V:

$$U = \frac{k + fa}{\sigma^2 - f^2}\frac{P_1}{\rho_0} \qquad (22)$$

$$V = \frac{fk + \ \ a}{\sigma^2 - f^2}\frac{P_1}{\rho_0}. \qquad (23)$$

Solving for the vertical pressure gradient:

$$\frac{dP_1}{dZ} = \frac{N^2 - \sigma^2}{\sigma}\rho_0 W \qquad (24)$$

where $N^2 = \frac{-g}{\rho_0}\frac{d\quad_0}{dZ}$ as previously defined. When one inserts (22) and (23) in (9):

$$\frac{P_1}{\rho_0} = -\frac{(\sigma^2 + f^2)}{\sigma(k^2 - a^2)}\frac{dW}{dZ}. \qquad (25)$$

When P_1 is eliminated from (24) and (25) and the transverse boundaries are disregarded (i.e., set $a = 0$),

$$\frac{d^2W}{dZ^2} + \left(\frac{1}{\rho_0}\frac{d\rho_0}{dZ}\right)\frac{dW}{dZ} + \left[\frac{k^2}{\sigma^2 - f^2}\right][N^2 - \sigma^2]W = 0 \qquad (26)$$

Equation (26) is the internal wave equation describing the response by a water column to a plane internal wave of known frequency, σ. The order of magnitude of the second term is small compared to the first and third. However, the numerical program can include the second term without any difficulty. It will be so retained.

B. APPLICATION OF RAY THEORY

An analytical solution to equation (26) may be made by the transformation:

$$W = \; (Z) \; \exp \left(\frac{1}{\rho_0} \frac{\partial \rho_0}{\partial Z} \frac{Z}{2} \right). \tag{27}$$

When the effect of the second term in eq. (26) is small, the analytical solution to the internal wave equation can, with a high degree of accuracy, be represented by Ψ, the oscillatory portion. The oscillatory solution is:

$$\Psi_{ZZ} + \left(\frac{k^2}{\sigma^2 - f^2} \right) (N^2 - \sigma^2) W = 0 \tag{28}$$

Solutions to equation (28) are of the form:

$$\Psi = A \; \exp \; i \left(k_1 X \pm k_1 \; \frac{N^2 - \;^2}{\sigma^2 - f^2} \; Y \pm \sigma t \right) \tag{29}$$

where k_1 = horizontal wave number.

Define Θ as the angle between the horizontal and the phase velocity. Then,

$$\cos \Theta = \frac{k_1}{\left(k_1^2 + k_1^2 \left[\frac{N^2 - \sigma^2}{\sigma^2 - f^2} \right] \right)^{1/2}} = \left[\frac{\sigma^2 - f^2}{N^2 - f^2} \right]^{1/2} \tag{30}$$

Since

$$(\sigma^2 - f^2)^{1/2} = (N^2 - \sigma^2)^{1/2} \; \frac{k_1}{(k_1^2 + k_2^2)^{1/2}} \tag{31}$$

$$\sigma = \left[\frac{k_1^2 N^2 + k_2^2 f^2}{k_1^2 + k_2^2} \right]^{1/2} \tag{31a}$$

where

$$k_2 = k_1 \left(\frac{N^2 - \;^2}{\sigma^2 - f^2} \right)^{1/2} \tag{32}$$

$$|k| = (k_1^2 + k_2^2)^{1/2} \tag{33}$$

Figure 2 illustrates the angle between the horizontal and the phase velocity. The phase velocity, broken into X and Z components, is:

$$C_X = \frac{k_1}{k_1^2 + k_2^2} = \frac{k_1}{(k_1^2 + k_2^2)^{3/2}} [k_1^2N^2 + k_2^2f^2]^{1/2} \tag{34}$$

and

$$C_Z = \frac{k_2}{(k_1^2 + k_2^2)^{3/2}} [k_1^2N^2 + k_2^2f^2]^{1/2} \tag{35}$$

For the group velocity,

$$\vec{C}_g = \nabla_k \sigma \tag{36}$$

where

$$\nabla_k = (\frac{\partial}{\partial k_1} \vec{i} + \frac{\partial}{\partial k_2} \vec{j}) \tag{37}$$

$$\vec{C}_g = \nabla_k [(\frac{k_1^2N^2 + k_2^2f^2}{k_1^2 + k_2^2})^{1/2}] \tag{38}$$

The X-component of group velocity is:

$$(C_g)_X = \frac{\partial}{\partial k_1}[(\frac{k_1^2N^2 + k_2^2f^2}{k_1^2 + k_2^2})^{1/2}]$$

$$= \frac{k_1 k_2^2}{|k|^3}[\frac{(N^2 - f^2)}{(k_1^2N^2 + k_2^2f^2)^{1/2}}] \tag{39}$$

The Z-component of group velocity is:

$$(Cg)_Z = \frac{\partial}{\partial k_2}[(\frac{k_1^2N^2 + k_2^2f^2}{k_1^2 + k_2^2})^{1/2}]$$

$$= -\frac{k_1^2 k_2}{|k|^3} [\frac{(N^2 - f^2)}{(k_1^2N^2 + k_2^2f^2)^{1/2}}] \tag{40}$$

Define Θ_g as the angle made with the horizontal by the group velocity vector. Then,



FIGURE 2. Illustration of Angle between Horizontal and Phase Velocity

$$\tan \Theta_g = \frac{(Cg)_Z}{(Cg)_X} = \frac{-K_1}{K_2} = \left[\frac{N^2 - f^2}{\sigma^2 - f^2}\right]^{1/2} \tag{41}$$

The ray trace part of the program drew the ray paths as depicted in equation (41) for ten frequencies selected within the inertio-gravity band.

Determining the dot product of the phase and group velocities:

$$\vec{C} \circ \vec{C_g} =$$

$$\left(\frac{k_1(k_1{}^2N^2 + k_2{}^2f^2)^{1/2}}{|k|^3}\right)\left(\frac{k_1k_1{}^2}{|k|^3}\ \frac{(N^2 - f^2)}{(k_1{}^2N^2 + k_2{}^2f^2)^{1/2}}\right)$$

$$- \left(\frac{k_2(k_1{}^2N^2 + k_2{}^2f^2)^{1/2}}{|k|^3}\right) \left(\frac{k_1{}^2k_2(N^2 - f^2)}{|k|^3(k_1{}^2N^2 + k_2{}^2f^2)^{1/2}}\right)$$

$$= 0 \tag{42}$$

Thus the phase velocity is perpendicular to the group velocity. Also,

$$C_Z + (C_g)_Z = \frac{k_2(k_2{}^2 - k_1{}^2)f^2}{(k_1{}^2N^2 + k_2{}^2f^2)^{1/2}(k_1{}^2 + k_2{}^2)^{3/2}} \tag{43}$$

The phase velocity and group velocity then have a relationship as seen in Figure 3.

C. NORMAL MODE APPLICATION

With the use of suitable boundary conditions, equation (26) may be solved to obtain modal structures. The applicable conditions are:

$$W = 0 \quad \text{at} \quad Z = 0 \tag{44}$$



Figure 3. Relation between Phase and Group Velocities

and

$$\frac{dW}{dZ} - \frac{k^2}{\sigma^2 - f^2} gW = 0 \quad \text{at} \quad Z = h \tag{44a}$$

To a high degree of approximation the vertical velocity shear is negligible at the surface and the second condition becomes:

$$W = 0 \quad \text{at} \quad Z = h \tag{45}$$

Equations (26), (44), and (45) compose a Sturmian system. W, the vertical velocity, is the eigenfunction. For a given frequency, discrete wave numbers or eigenvalues are determined. The program utilized a fourth-order Runge-Kutta numerical method to obtain an end value after integration of the internal wave equation over the water column. The endpoint was refined using a Newton-Raphson iteration procedure.

III. DEVELOPMENT OF THE MODEL

A. MAIN PROGRAM

The main program prints the station density stratification. The program then calls upon the three sub-programs to interpolate density between station depths, to plot ray paths for selected internal wave frequencies, and to print out and plot modal structures for these frequencies.

B. SUBROUTINE INTERP

Subroutine INTERP interpolates the density stratification for 400 incremental depths through the water column. The interpolation is accomplished by the use of a third-degree polynomial that matches points and slopes at each depth where data was obtained from a Nansen bottle. To guarantee density stability, the smaller of the slope above or below each point is chosen as the slope at that point. From the interpolation polynomial, INTERP computes the Vaisala frequency for each density value.

C. SUBROUTINE RAY

Subroutine RAY plots the interpolated density and the Vaisala frequency as a function of depth. RAY then determines the inertio-gravity frequencies at rates of 1/100, 1/10, 2/10, 3/10, 4/10, 5/10, 6/10, 7/10, 8/10, and 9/10 of the frequency band. The program then plots the characteristic paths over the depth of the column for each of these frequencies.

Additionally, subroutine RAY determines the percentage of the water column depth where the selected frequency exceeds the Vaisala frequency. This percentage represents that portion of the column where the Sturmian solution to the internal wave equation is exponential. A greater percentage implies increasing difficulty in applying boundary conditions during the numerical integration. The percentage figure for each frequency is listed beneath the ray path plot.

D. SUBROUTINE RUNK

Subroutine RUNK utilizes a fourth-order Runge-Kutta numerical method to integrate the internal wave equation over the water column. The endpoint of the integration is refined by a Newton-Raphson iteration method until the endpoint is 1/200 of the maximum value of the eigenfunction, W, determined during the integration scheme. For each selected frequency, the first approximation for the initial eigenvalue is

$$K_1 = \frac{\sigma}{\sqrt{gh}} \tag{45}$$

The approximation is refined until the wave number corresponding to the first mode is determined. The approximation for the second mode is twice the value of K_1. The initial estimate of the remaining eight eigenvalues is the previous eigenvalue plus the difference between the previous two eigenvalues.

After the first three eigenvalues are determined, the program plots their modal structure. The remaining seven wave numbers are then listed.

As the selected frequencies within the inertio-gravity band increase in magnitude, the boundary constraints, though analytically well-defined, weaken exponentially. The higher frequencies are greater than the incremental Vaisala frequency for a greater percentage of column depth and it becomes increasingly difficult for the numerical program to find eigenvalues. RUNK terminates its search for the next eigenvalues by (1) stopping after 20 refinements by the Newton-Raphson iteration process, (2) restricting the magnitude of the eigenfunction to 10^{20} cm/sec, and (3) stopping if the solution converges to a mode other than the one searched for.

Also, as the frequency within the band increases, W becomes more sensitive to changes in k when seeking higher modes. For situations where a minute change in k causes abrupt responses in end values of W, the eigenvalue may be only refined to a certain degree. These eigenvalues are indicated as approximations.

IV. TESTING THE PROGRAM

A. LINEAR INPUT

To check numerical formulation, coding, and accuracy, two test cases were used whose analytic solutions were possible. The first test was a linear variation of depth from 1.023 grams/cc at the surface to 1.033 grams/cc at the bottom depth of 1000 meters. The Vaisala frequency for this input was constant and equal to:

$$N = \sqrt{\frac{-g}{\rho}\frac{\partial\rho}{\partial z}}$$

$$= \sqrt{\frac{9.8\ \text{M/sec}^2}{1.025\ 614/\text{cc}} \times 10^{-5}} = 9.77 \times 10^{-3}\,\text{sec}^{-1}. \qquad (46)$$

The plot by the program of the density matches that of the linear variation. The Vaisala frequency determined numerically matches that determined analytically. Table I compares the analytical and numerical results of the linear variation of density for frequency 0.00684 cycles/min. Figure 4 compares the first three modal structures. Analytically, the normal modes are

$$K_m = \left(\frac{\sigma^2 - f^2}{N^2 - f^2}\right)^{1/2} \left(\frac{m\pi}{h}\right). \qquad (47)$$

The eigenfunction is:

$$\exp\left(\frac{N^2}{g}\frac{Z}{2}\right)\ \sin\left(\frac{m\pi}{h}Z\right) \qquad (48)$$

B. THREE-LAYER INPUT

The second input was a density strata that was constant at 1.023 grams/cc from the surface to 300 meters, varied

Table I. A comparison of the first ten normal modes determined for frequency 0.00684 cycles/min:

	Analytical	Numerical	% Error
k_1	3.613×10^{-5}	3.629×10^{-5}	0.45
k_2	7.225×10^{-5}	7.259×10^{-5}	0.47
k_3	1.083×10^{-4}	1.088×10^{-4}	0.27
k_4	1.445×10^{-4}	1.451×10^{-4}	0.42
k_5	1.806×10^{-4}	1.814×10^{-4}	0.44
k_6	2.167×10^{-4}	2.177×10^{-4}	0.46
k_7	2.529×10^{-4}	2.540×10^{-4}	0.43
k_8	2.890×10^{-4}	2.903×10^{-4}	0.45
k_9	3.251×10^{-4}	3.266×10^{-4}	0.46
k_{10}	3.613×10^{-4}	3.629×10^{-4}	0.44



FIGURE 4. Comparison of Eigenfunctions for First 3 Modes of Frequency 0.00684

linearly from 300 meters to a value of 1.027 grams/cc at 700 meters, and then remained invariant from 700 meters to the bottom at a depth of 1000 meters. The corresponding value of the Vaisala frequency was then zero in the upper and lower isopycnal layers and equal to 9.8×10^{-3} in the middle region.

The density plot by the program matched the analytic solution to the third decimal place. Figure 5 illustrates the comparison between the analytic solution and the computer solution for the Vaisala frequency, and illustrates how the interpolation scheme handles discontinuities in the density slope.



FIGURE 5. Comparison of Vaisala Frequency Plots by Analytical and Numerical Methods

V. APPLICATION OF THE PROGRAM TO OCEANOGRAPHIC DATA

The program utilizes density strata from Nansen casts. It can process several stations. IDATA, the number of stations to be studied, is the first card to be typed. It follows the alphanumerics that are read in and serves in the plotting portions of the program. IDATA is read in as an integer, right-oriented to column ten of the IBM card. The second card to be typed up consists of three figures. The first is M, the number of bottles at the station, read in as an integer, right-oriented to column 10. The second is ZMAX, the depth at which the last bottle was taken, read in the format F10.1, right-oriented to column 20. The third figure is RLAT, the latitude of the station, that is read in using format F10.1 right-oriented to column 30.

The next cards to be typed are those representing the density strata. These are read in as four pairs of numbers for each card. Each number has a field width of ten and is right-oriented. The first number in each pair represents station depth and has a format of F10.1. The second of the pair has a format F10.5 and represents the station density. These four pairs of numbers, representing 80 columns per card, are entered until all the station data has been read in.

The above represent all the cards needed to be typed for each station. From experience gained using the IBM 360 computer to process ocean stations taken in mid-August, 1970,

off Monterey, California, it appears that two minutes are required to process shallower stations (around 200 meters) and four minutes to process deeper stations (around 2000 meters).

The actual printed output for each station appears as follows. On the first page of printed output appears: (1) the number of station depths, (2) the lowest bottle depth, (3) a printout of the station depths and densities in columnar form, (4) the inertial frequency computed for the station, (5) the maximum Vaisala frequency computed for the station, and (6) the ten frequencies within the inertio-gravity band selected for consideration.

The second page of printed output plots the Vaisala frequency and density versus depth.

The next ten pages plot the ray path for each frequency versus depth. At the bottom of each ray plot, the amount of time that the frequency is greater than the Vaisala frequency is indicated.

The remaining twenty pages of output display information on normal modes. Corresponding to each of the ten selected frequencies are two pages. The first page lists the first three normal modes and plots these modes versus depth. The second page lists the remaining seven modes.

Two stations, one shallow and one deep, were chosen to illustrate the output from the program. The results for both stations are summarized by the following tables and figures.

Table II lists the information for the deep oceanographic station displayed on the first page of computer output. This station is representative of an oceanic internal wave regime. Figure 6 shows the plot of the density variation and the Vaisala frequency for the deep station. The Vaisala plot follows the pattern of the density profile.

Figure 7 depicts the ray paths for the first three selected frequencies. The first frequency selected, 0.01371 cycles/min, never exceeds the incremental Vaisala frequency. Thus, theoretically, its solution is always oscillatory and it travels horizontally nearly 30 km. The second frequency selected, 0.12969 cycles/min, exceeds the Vaisala frequency in the region of the water column below a depth of 970 meters. Thus it exponentially 'decays' in this region and its ray path over this portion of the water column is vertical. Figures 7 and 8 illustrate that, with increasing frequency, the horizontal distance covered by the ray path decreases and a greater portion is in the exponential region. The ray paths for the last four selected frequencies were not shown since their ray is nearly vertical from top to bottom.

Figure 9 plots the first three modes for frequency 0.01371 cycles/min. This was the only frequency for which the first three mornal modes could be positively identified. Besides gaining a better understanding of the actual appearance of the modal structures, Figure 9 suggests optimum placement of a thermocline follower for mode detection. For example, the follower would best detect mode three at a depth of about

TABLE II. Data display for deep station.

Incremental depth	Incremental density
0.0	1.02498
10.0	1.02505
20.0	1.02522
30.0	1.02565
50.0	1.02588
75.0	1.02611
100.0	1.02624
150.0	1.02646
200.0	1.02656
300.0	1.02679
400.0	1.02681
500.0	1.02702
600.0	1.02711
800.0	1.02730
1000.0	1.02743
1200.0	1.02750
1500.0	1.02759
2000.0	1.02770
2500.0	1.02774
3000.0	1.02776

Frequency number	Cycles/min
inertial	0.00082
1	0.01371
2	0.12969
3	0.25855
4	0.38741
5	0.51628
6	0.64514
7	0.77401
8	0.90287
9	1.03173
10	1.16059
maximum Vaisala	1.28946



FIGURE 6

FIGURE 6. Density variation and Vaisala frequency for deep station



FIGURE 7

FIGURE 7. Ray paths for deep station



FIGURE 8. Ray paths for third through sixth frequency at deep station.



FIGURE 9. Modal structures for first three modes of first frequency

2000 meters. Table III lists the various modes determined for the selected frequencies for the deep oceanographic station.

Table IV displays the first page of printed output for the shallow station representative of the regime in a large bay. Figure 10 illustrates the density variation and the Vaisala frequency as a function of depth. As in the deep station, the first selected frequency of the shallow station has an oscillatory solution throughout most of the water column. Its ray path travels horizontally nearly 2500 meters. Its exponential region may be seen as a vertical path at a distance of 1955 meters. Figure 11 shows the effect of an increase in frequency without a considerable increase in the effect of exponential decay. The second frequency selected, 0.15818 cycles/min., has an exponential solution over the water column in nearly the same region as the first. However, frequency 2 only travels a horizontal distance of 230 meters. Figure 12 depicts the decrease in ray path as the frequency increases.

Figure 13 depicts the first three normal modes for frequency 0.01656 cycles/min. Figure 14 illustrates the first three normal modes for frequency 0.31554 cycles/min. Figure 14 illustrates well the manner in which the modal structure deteriorates with increased frequency. Table V lists the modes determined for selected frequencies at the shallow station.

Table III. Normal modes for selected frequencies at the deep station.

0.01371 cycles/min.		0.12969 cycles/min.	
1.	0.00011736	1.	0.0018469
2.	0.00023344	2.	0.0034051
3.	0.00032185		
4.	0.00043948		
5.	0.00055503		
6.	0.00063828		
7.	0.00079211		
8.	0.00085791		
9.	0.00096286		
10.	0.00011641		

Table IV. Data display for shallow station.

Incremental depth	Incremental density
0.0	1.02535
5.0	1.02537
10.0	1.02543
15.0	1.02551
20.0	1.02570
25.0	1.02580
30.0	1.02606
40.0	1.02610
50.0	1.02614
60.0	1.02614
75.0	1.02616
100.0	1.02619

Frequency number	Cycles/min
inertial	0.00083
1	0.01656
2	0.15818
3	0.31554
4	0.47289
5	0.63024
6	0.78760
7	0.94495
8	1.10231
9	1.25966
10	1.41702
maximum Vaisala	1.57437



FIGURE 10. Density profile and Vaisala frequency at the shallow station



FIGURE 11. Ray plots for first two selected frequencies at the shallow station.



FIGURE 12. Ray plots for third through ninth frequencies at the shallow station



FIGURE 13. Modal structure for first selected frequency at shallow station



FIGURE 14. Modal structures for third selected frequency at shallow station

Table V. Normal modes for selected frequencies at the shallow station.

0.01656 cycles/min.

1. 0.00087701
2. 0.0025324
3. 0.0035674
4. 0.0050806
5. 0.0055217

0.31554 cycles/min.

1. 0.018642
2. 0.056013
3. 0.10871
4. 0.14614
5. 0.17507
6. 0.20301
7. 0.23839
8. 0.28283
9. 0.33293
10. 0.37399

0.47289 cycles/min.

1. 0.032306
2. 0.094660
3. 0.19395
4. 0.26040
5. 0.33316
6. 0.40811
7. 0.48565
8. 0.55884
9. 0.63999
10. 0.72820

0.15818 cycles/min.

1. 0.0086086
2. 0.025461
3. 0.041537
4. 0.051653
5. 0.069448
6. 0.074546
7. 0.092209
8. 0.10860
9. 0.12385
10. 0.13605

0.63024 cycles/min.

1. 0.053539
2. 0.14805

0.78760 cycles/min.

1. 0.092463
2. 0.23132

VI. CONCLUSIONS

The numerical program developed in this thesis has derived a technique that links the theoretical application of internal waves to real data. Real solutions are determined for actual density profiles. The program can yield a useful description of the internal wave regime within a given area. From representative samples of stations investigated, four points are apparent:

(1), that with greater depth the effect of the boundary conditions weakens;

(2), that for low frequencies that exceed the Vaisala frequency over only a small percentage of the water column, the ray path approaches the horizontal, while for higher frequencies the ray path approaches the vertical;

(3), that the plot of the lowest three modes demonstrates the manner in which, with increasing frequency, the effect of the boundary conditions weakens as the oscillatory regions give way to exponential "decay" regions;

(4), that for the stations investigated by this program, only those frequencies within the lower quarter of the inertio-gravity frequency band could be solved for normal modes with any degree of success.

APPENDIX

```
C ***********************************************************************
C *                                                                     *
C *                                                                     *
C *  THE MAIN PROGRAM SELECTS TEN FREQUENCIES WITHIN A BAND OF FREQUEN-*
C *  CIES BOUNDED ABOVE BY THE MAXIMUM VAISALA FREQUENCY WITHIN THE    *
C *  COLUMN AND BELOW BY THE INERTIAL FREQUENCY. FOR EACH SELECTED     *
C *  FREQUENCY THE PROGRAM PLOTS THE RAY PATH OF PROPAGATING ENERGY.   *
C *  ALSO THE PROGRAM DETERMINES THE FIRST TEN NORMAL MODES ASSOCIATED *
C *  WITH EACH FREQUENCY.                                              *
C *                                                                     *
C *                                                                     *
C ***********************************************************************
      COMMON M,RLAT,G,Z(30),RHOSTA(30),RHO(400),ZMAX,VASQRT,VAISAL(400),
     *BLANK,DOT,O,E,STAR,FREQ(10),DEPTH(400),OMEGA,AMP,ONE,TWO,THREE
C
C THE FOLLOWING ALPHANUMERICS TO BE READ IN ARE USED IN THE PLOTTING
C PROCEDURES.
      READ(5,60) BLANK,DOT,STAR,O,E,AMP,ONE,TWO,THREE
   60 FORMAT (3A1,2A4,4A1)
C
C IDATA REPRESENTS THE NUMBER OF STATIONS FROM WHICH DATA HAS BEEN
C GATHERED.
      READ (5,10)IDATA
   10 FORMAT (I10)
C
C G IS THE GRAVITATIONAL ACCELERATION.
      G= 9.8
C
C     ZMAX IS THE WATER DEPTH.
C     M IS THE NUMBER OF STATIONS.
C RLAT IS THE VALUE OF LATITUDE IN DEGREES AND WILL BE USED TO COMPUTE
C THE INERTIAL FREQUENCY AT THE STATION LOCATION.
   20 READ (5,30) M,ZMAX,RLAT
   30 FORMAT(I10,F10.1,F10.2)
      RLAT = RLAT/57.6
      WRITE(6,40) M,ZMAX
   40 FORMAT('1',10X,'THE NUMBER OF INCREMENTAL DEPTHS AT WHICH DATA WAS
     * OBTAINED AT THIS STATION ARE ',I2,'.',///,10X,'THE LOWEST DEPTH
     *AT WHICH DATA WAS TAKEN IS ',F6.1,' METERS. THE DENSITY STRATA IS
     *AS FOLLOWS:',///,43X,'INCREMENTAL DEPTH',10X,'INCREMENTAL DENSITY'
     *,/)
C
C     THE STATION DATA IS NOW READ IN.
C     Z WILL IDENTIFY THE DEPTHS AT WHICH THE STATION DATA WAS GATHERED.
C     RHOSTA IS THE DENSITY COMPUTED FOR THAT STATION.
      READ(5,103) (Z(I),RHOSTA(I), I=1,M)
  103 FORMAT (4(F10.1,F10.5))
C     ECHO CHECK TO ENSURE THAT Z AND RHOSTA WERE ENTERED CORRECTLY.
```

```
      WRITE(6,104) (Z(I),RHOSTA(I), I=1,M)
  104 FORMAT(48X,F7.1,19X,F7.5,/)
      CALL INTERP
      CALL RAY
      CALL RUNK
C THE PROGRAM CONTINUES TO COMPUTE THE INTERNAL WAVE CHARACTERISTICS
C UNTIL ALL STATION DATA HAVE BEEN READ.
      IDATA=IDATA-1
      IF (IDATA.GT.0) GO TO 20
   50 STOP
      END
```

```
      SUBROUTINE INTERP
C *****************************************************************
C *                                                               *
C *  SUBROUTINE INTERP USES A CUBIC EQUATION TO INTERPOLATE THE DENSITY*
C *  STRATIFICATION THROUGH 400 POINTS. THE PROGRAM ALSO COMPUTES THE  *
C *  VAISALA FREQUENCY FOR EACH INCREMENTAL DEPTH.                     *
C *****************************************************************
      COMMON M,RLAT,G,Z(30),RHOSTA(30),RHO(400),ZMAX,VASQRT,VAISAL(400),
     *BLANK,DOT,O,E,STAR,FREQ(10),DEPTH(400),OMEGA,AMP,ONE,TWO,THREE
      DIMENSION STORE(30),D(30)
      D(1)=(RHOSTA(2)-RHOSTA(1))/(Z(2)-Z(1))
      STORE(1)=D(1)
      MM = M-1
      DO 110 I=2,MM
      D(I)=(RHOSTA(I+1)-RHOSTA(I))/(Z(I+1)-Z(I))
      STORE(I)=D(I)
  110 IF(D(I).GT.STORE(I-1)) D(I)=STORE(I-1)
      D(M) = D(MM)
      K=1
      N=1
      RHO(1)=RHOSTA(1)
      DEPTH(1)=0.
      DELZ = ZMAX/399.
      VAISM = 0.
      DO 120 K=2,400
      DEPTH(K) = DELZ*(K-1)
  130 IF (DEPTH(K).GT.Z(N)) GO TO 140
      GO TO 150
  140 N=N+1
      GO TO 130
  150 ZZ=(DEPTH(K)-Z(N-1))/(Z(N)-Z(N-1))
      A = RHOSTA(N-1)
      B = RHOSTA(N)
      C = D(N-1)*(Z(N)-Z(N-1))
      F = D(N)*(Z(N)-Z(N-1))
      C1 = 2.*A-2.*B+C+F
      C2 = -3.*A+3.*B-2.*C-F
      C3 = C
      C4 = A
      RHO(K) = ((C1*ZZ+C2)*ZZ+C3)*ZZ+C4
      VAISAL(K) = (G/RHO(K))*((3.*C1*ZZ+2.*C2)*ZZ+C3)/(Z(N)-Z(N-1))
  120 IF (VAISAL(K).GT.VAISM) VAISM = VAISAL(K)
      VAISAL(1) = VAISAL(2)
      VASQRT = SQRT(VAISM)
      RETURN
      END
```

```
      SUBROUTINE RAY
C
C ************************************************************************
C *                                                                      *
C *                                                                      *
C *   SUBROUTINE RAY COMPUTES THE RANGE OF FREQUENCIES OF INTERNAL WAVES*
C *   THAT MAY EXIST WITHIN THE WATER COLUMN AND USING A GRAPHING TECH- *
C *   NIQUE PLOTS A ROUGH PROFILE OF THE RAY PATHS OF TEN FREQUENCIES   *
C *   THAT ARE EVENLY SPACED THROUGHOUT THE FREQUENCY SPECTRUM. THE     *
C *   UPPER BOUND IS THE MAXIMUM VAISALA FREQUENCY AND THE LOWER BOUND  *
C *   IS THE INERTIAL FREQUENCY.                                        *
C *        THE PROGRAM ALSO DETERMINES THE PER CENTAGE OF TIME THAT THE *
C *   SELECTED FREQUENCY IS GREATER THAN THE VAISALA FREQUENCY AT       *
C *   EACH INCREMENTAL DEPTH. THE PER CENTAGE IS PRINTED OUT FOR EACH   *
C *   FREQUENCY AND SERVES AS AN INDICATION OF THE RELATIVE DIFFICULTY  *
C *   WITH WHICH THE NORMAL MODES MAY BE OBTAINED.                      *
C *                                                                      *
C *                                                                      *
C ************************************************************************
C
      REAL LINE
      COMMON M,RLAT,G,Z(30),RHOSTA(30),RHO(400),ZMAX,VASQRT,VAISAL(400),
     *BLANK,DOT,O,E,STAR,FREQ(10),DEPTH(400),OMEGA,AMP,ONE,TWO,THREE
      DIMENSION RATIO(400),LINE(100),S(400),S1(80),VAIS(400)
C OMEGA IS THE INERTIAL FREQUENCY
      OMEGA = SIN(RLAT)/720.
      VASMIN = VASQRT*60.
      WRITE(6,5) OMEGA,VASMIN
    5 FORMAT(10X,'THE INERTIAL FREQUENCY IS COMPUTED TO BE ',F8.5,' CYCL
     *ES/MIN.',//,10X,'THE MAXIMUM VAISALA FREQUENCY IS COMPUTED TO BE '
     *,F8.5,' CYCLES/MIN.',//)
      WRITE (6,1)
    1 FORMAT (10X,'THE FREQUENCIES TO BE CONSIDERED ARE:',//)
C RANGE IS CREATED SO THAT WE CONSIDER ONLY THOSE FREQUENCIES THAT ARE
C BOUNDED BY THE VAISALA FREQUENCY  AT A MAXIMUM WITHIN THE WATER COLUMN
C AND BY THE INERTIAL FREQUENCY AS A MINIMUM.
      RANGE = (VASMIN - OMEGA)/10.
      FREQ(1) = RANGE/10. + OMEGA
      WRITE(6,2) FREQ(1)
    2 FORMAT(10X,'FREQUENCY NO.  1 IS ',F8.5,' CYCLES/MIN.',/)
      FREQ(1) = FREQ(1)/60.
      DO 3 I=2,10
      FREQ(I) = RANGE*(I-1) + OMEGA
      WRITE(6,4) I,FREQ(I)
    4 FORMAT(10X,'FREQUENCY NO. ',I2,' IS ',F8.5,' CYCLES/MIN.',/)
    3 FREQ(I) = FREQ(I)/60.
```

```
      DO 199 M=1,400
      IF (VAISAL(M).LT.0.) VAISAL(M) = 0.
  199 VAIS(M)=SQRT(VAISAL(M))
      DO 11 J = 1,100
   11 LINE(J) = DOT
      WRITE(6,16) LINE
   16 FORMAT('1','DEPTH(M) VAISALA FREQ. DENSITY',2X,100A1)
      DO 21 N=2,100
   21 LINE(N) = BLANK
      DO 51 I=1,400,5
      LINE(1)=DOT
      FACTOR = (VAIS(I)/VASQRT)*100.
      K = IFIX(FACTOR)
      IF (K.LT.1) K=1
      IF (K.GT.100) K=100
      LINE(K)=STAR
      STEP = ((RHO(I)-RHO(1))/(RHO(400)-RHO(1)))*100.
      J = IFIX(STEP)
      IF (J.LT.1) J=1
      IF (J.GT.100) J=100
      LINE(J) = AMP
      WRITE (6,31) DEPTH(I),VAIS(I),RHO(I),LINE
   31 FORMAT(2X,F6.1,1X,E13.4,3X,F7.5,1X,100A1)
      LINE(K) = BLANK
      LINE(J) = BLANK
   51 CONTINUE
      DO 15 I=1,10
      SIGMA = FREQ(I)
      SG = SIGMA**2
      DO 65 J=2,100
   65 LINE(J)=BLANK
      SUM=0.
      N=0
      DELZ=ZMAX/400
      DO 40 M=1,400
      IF (SG.GT.VAISAL(M)) GO TO 20
   10 RATIO(M) = (VAISAL(M)-SG)/(SG-(OMEGA/60.)**2)
      GO TO 30
   20 RATIO(M)=0.
      N=N+1
   30 ENTRY=SQRT(RATIO(M))*DELZ
      SUM=SUM+ENTRY
   40 S(M) = SUM
      RN=N
      PRCENT=RN/4.
      WRITE(6,70) SUM
   70 FORMAT('1',1X,'DEPTH(M)  EFFECT  HOR. DIST.(M)',84X,F7.1,' METERS'
     *)
```

```
      DO 90 K=1,400,5
      KK = 1 + K/5
      S1(KK) = S(K)
      IF (SUM.EQ.0.) GO TO 200
      S(K) = (S(K)/SUM)*100.
      GO TO 201
  200 S(K) = 0.
  201 J = IFIX(S(K))
      IF (J.LT.1) J=1
      IF (J.GT.100) J = 100
      LINE(1) = DOT
      LINE(25) = DOT
      LINE(50) = DOT
      LINE(75) = DOT
      LINE(100) = DOT
      LINE(J) = STAR
      IF (SG.GT.VAISAL(K)) GO TO 25
      IF (SG.LE.VAISAL(K)) GO TO 35
   35 WRITE(6,100) DEPTH(K),O,S1(KK),LINE
      GO TO 45
   25 WRITE (6,100) DEPTH(K),E,S1(KK),LINE
  100 FORMAT(1X,F7.2,5X,A4,2X,F7.1,3X,100A1)
   45 LINE(J) = BLANK
   90 CONTINUE
      SIG = SIGMA*60.
      WRITE (6,50) SIG,PRCENT
   50 FORMAT(' ','FREQUENCY',F8.5,' IS GREATER THAN THE VAISALA FREQUENC
     *Y ',F4.1,'% OF THE TIME.')
   15 CONTINUE
      RETURN
      END
```

```
      SUBROUTINE RUNK
C
C ***********************************************************************
C *                                                                     *
C *                                                                     *
C *   SUBROUTINE RUNK TAKES THE VARIABLES PREVIOUSLY CREATED AND UTILI- *
C *   ZING A FOURTH ORDER RUNGE-KUTTA METHOD SOLVES THE SECOND ORDER    *
C *   DIFFERENTIAL EQUATION GOVERNING THE VERTICAL VELOCITY ASSOCIATED  *
C *   WITH THE PASSAGE OF AN INTERNAL WAVE. THE FORM OF THE EQUATION IS:*
C *                                                                     *
C *   W" = -PHI*W' - (K2/(S2-F2))*(N2-S2)*W                             *
C *                                                                     *
C *   WHERE:                                                            *
C *                                                                     *
C *   W IS THE VERTICAL VELOCITY IN THE WATER COLUMN                    *
C *   PHI IS THE DERIVATIVE OF DENSITY WRT DEPTH.                       *
C *   K2 IS THE SQUARE OF THE WAVE NUMBER.                              *
C *   S2 IS THE SQUARE OF THE FREQUENCY.                                *
C *   F2 IS THE INERTIAL FREQUENCY SQUARED.                             *
C *   N2 IS THE VAISALA FREQUENCY SQUARED.                              *
C *                                                                     *
C *   FOR EACH OF THE TEN SELECTED FREQUENCIES THE PROGRAM ATTEMPTS TO  *
C *   COMPUTE THE FIRST TEN MODES WHICH REPRESENT THE FIRST TEN EIGEN   *
C *   VALUES ASSOCIATED WITH THE DIFFERENTIAL EQUATION.                 *
C *                                                                     *
C *                                                                     *
C ***********************************************************************
C
      COMMON M,RLAT,G,Z(30),RHOSTA(30),RHO(400),ZMAX,VASQRT,VAISAL(400),
     *BLANK,DOT,O,E,STAR,FREQ(10),DEPTH(400),OMEGA,AMP,ONE,TWO,THREE
      DIMENSION W(201),GUESS(11),WONE(67),WTWO(67),WTHREE(67),GRAPH1(24)
     *,GRAPH2(49),GRAPH3(49),PHI(400)
C SINCE THE Z-AXIS IS UP IN FELDSTAD'S EQUATION THE SENSE OF PHI MUST BE
C SUCH THAT A RATE OF DENSITY INCREASE DOWNWARD ASSUMES A NEGATIVE VALUE
      DO 219 I=1,400
  219 PHI(I)=-VAISAL(I)/G
      DO 10 II = 1,10
      SIGMA = FREQ(II)
      IF (SIGMA.EQ.0.) GO TO 10
      SIG = SIGMA*60.
      WRITE (6,150) SIG
  150 FORMAT ('1',20X,'COMPUTATION OF FIRST THREE MODES FOR FREQUENCY ',
     *F8.5,' CYCLES/MIN.',/)
      SG=SIGMA**2
```

```
      CAY=SIGMA/SQRT(G*ZMAX)
      OMEGA = CMEGA/60.
      BIG = 10.**20
      EPS = 0.
      NN = 1
      JJ = 1
      L=1
      MM = 1
   20 LL=1
  590 Z1=0.
      W1=0.0
      F1= -.0001
      S1=0.
      H=-ZMAX/200.
      FEL = CAY**2/(SG-OMEGA**2)
      DH=H/2.
      W(1)=0.
      DO 300 I=1,399,2
      Z2=Z1+DH
      W2=W1+F1*DH
      F2=F1+S1*DH
      CON=PHI(I)
      S2=-CON*F2+FEL*(G*CON+SG)*W2
      W3=W1+F2*DH
      F3=F1+S2*DH
      S3=-CON*F3+FEL*(G*CON+SG)*W3
      W4=W1+F3*H
      F4=F1+S3*H
      CON=PHI(I+1)
      S4=-CON*F4+FEL*(G*CON+SG)*W4
      DY=H*(F1+2.*F2+2.*F3+F4)/6.
      DF=H*(S1+2.*S2+2.*S3+S4)/6.
      Z1=Z1+H
      W1=W1+DY
      F1=F1+DF
      S1=-CON*F1+FEL*(G*CON+SG)*W1
      J=(I+3)/2
      IF (ABS(W1).GT.BIG) GO TO 309
      IF(ABS(W1).GT.EPS) EPS = ABS(W1)
  300 W(J)=W1
      EPS = .005*EPS
      ENDVAL=W(201)
      KK=1
      DO 301 M=1,199
  301 IF (W(M)*W(M+1).LT.0.) KK=KK+1
      IF (ABS(ENDVAL).LT.EPS) GO TO 350
  315 IF(LL.EQ.1) GO TO 330
  306 IF (ABS(W(201)).GT.ABS(W(200))) GO TO 308
```

```
307 GO TO 316
308 IF (KK.GT.1) GO TO 316
    CAY = CAY*1.5
    GO TO 20
316 IF (STORE1.EQ.ENDVAL) GO TO 360
320 CAY=CAY-ENDVAL*(CAY2-CAY1)/(ENDVAL-STORE1)
    CAY1=CAY2
    CAY2=CAY
322 STORE1 = ENDVAL
    LL=LL+1
    IF (LL.EQ.21) GO TO 340
325 GO TO 590
330 STORE1=ENDVAL
    CAY1=CAY
    CAY = CAY*(1. + 0.1/L)
    CAY2 = CAY
    LL=LL+1
    GO TO 590
340 WRITE (6,345)
345 FORMAT(10X,' THE PROGRAM FAILS TO CONVERGE IN 20 ITERATIONS AND IS
   * PROCEEDING ON TO THE NEXT FREQUENCY.',/)
    GO TO 10
360 WRITE (6,365) L,CAY
365 FORMAT(10X,' VALUE OF MODE ',I2,' IS ',E14.7,' . APPROXIMATE VALUE
   *. EIGENFUNCTION NOT DETERMINED.',/)
    GO TO 378
350 M=1
    A=1
    B = 1
    DO  1 N=1,201
    IF (ABS(W(N)).LE.EPS) GO TO 1
    B = W(N)/ABS(W(N))
    IF (ABS(A-B).LT.0.5) GO TO 1
    M=M+1
    A = -A
  1 CONTINUE
    IF(MM.EQ.2) GO TO 370
    IF(M.EQ.1) GO TO 400
    CAY = CAY/M
    GO TO 20
400 CAY = ABS(CAY)
    WRITE (6,410) CAY
410 FORMAT (10X,' VALUE OF MODE  1 IS ',E14.7,' .',/)
    DO 415 IM = 1,201,3
    IN = 1 + IM/3
415 WONE(IN) = W(IM)
    GUESS(1) = CAY
    GUESS(2)=2.*CAY
```

```
      MM=2
  110 L=L+1
   95 CAY = GUESS(L)
      GO TO 20
  370 IF (M.LT.L) GO TO 115
      IF (M.GT.L) GO TO 130
      CAY = ABS(CAY)
      WRITE(6,375) M,CAY
  375 FORMAT (10X,' VALUE CF MODE ',I2,' IS ',E14.7,' .',/)
      IF (L.EQ.2) GO TO 376
      IF (L.EQ.3) GO TO 377
      GO TO 378
  376 DO 374 IM = 1,201,3
      IN = 1 + IM/3
  374 WTWO(IN) = W(IM)
      GO TO 378
  377 DO 373 IM = 1,201,3
      IN = 1 + IM/3
  373 WTHREE(IN) = W(IM)
      WRITE (6,379)
  379 FORMAT(' ',1X,'DEPTH(M)',8X,'MODE ONE',28X,'MODE TWO',39X,'MODE TH
     *REE',/)
      WMAX1 = WONE(1)
      WMAX2 = WTWO(1)
      WMAX3 = WTHREE(1)
      WMIN2 = WMAX2
      WMIN3 = WMAX3
      DO 380 N1 = 1,67
      IF (WONE(N1).GT.WMAX1) WMAX1 = WONE(N1)
      IF (WTWO(N1).GT.WMAX2) WMAX2 = WTWO(N1)
      IF (WTWO(N1).LT.WMIN2) WMIN2 = WTWO(N1)
      IF (WTHREE(N1).GT.WMAX3) WMAX3 = WTHREE(N1)
      IF (WTHREE(N1).LT.WMIN3) WMIN3 = WTHREE(N1)
  380 CONTINUE
      DIFF2 = WMAX2 - WMIN2
      DIFF3 = WMAX3 - WMIN3
      DO 391 NM = 1,24
  391 GRAPH1(NM) = BLANK
      DO 392 NL = 1,49
      GRAPH2(NL) = BLANK
  392 GRAPH3(NL) = BLANK
      DO 390 N2 = 1,67
      STEP1 = (WONE(N2)/WMAX1)*24.
      L1 = IFIX(STEP1)
      IF (L1.LT.1) L1 = 1
      GRAPH1(1) = DOT
      GRAPH1(L1) = ONE
       IF (WTWO(N2).GT.0.) WTWO(N2) = ABS(WMIN2) + WTWO(N2)
```

```
      IF (WTWO(N2).LE.0.) WTWO(N2) = ABS(WMIN2) - ABS(WTWO(N2))
      STEP2 = (WTWO(N2)/DIFF2)*49.
      L2 = IFIX(STEP2)
      IF (L2.LT.1) L2 = 1
      IF (N2.EQ.1) K2 = L2
      GRAPH2(K2) = DOT
      GRAPH2(L2) = TWO
      IF (WTHREE(N2).GT.0.) WTHREE(N2) = ABS(WMIN3)  + WTHREE(N2)
      IF (WTHREE(N2).LE.0.) WTHREE(N2) = ABS(WMIN3) - ABS(WTHREE(N2))
      STEP3 = (WTHREE(N2)/DIFF3)*49.
      L3 = IFIX(STEP3)
      IF(L3.LT.1) L3 = 1
      IF (N2.EQ.1) K3 = L3
      GRAPH3(K3) = DOT
      GRAPH3(L3) = THREE
      N3 = 6*N2 - 2
      WRITE (6,395) DEPTH(N3),GRAPH1,GRAPH2,GRAPH3
395   FORMAT(1X,F6.1,1X,24A1,1X,49A1,1X,49A1)
      GRAPH1(L1) = BLANK
      GRAPH2(L2) = BLANK
      GRAPH3(L3) = BLANK
390   CONTINUE
      WRITE(6,393)
393   FORMAT('1',20X,'COMPUTATION OF REMAINING SEVEN MODES.')
378   GUESS(L) = CAY
      GUESS(L+1) = 2.*CAY-GUESS(L-1)
      IF (L.EQ.10) GO TO 10
105   GO TO 110
115   IF (L.EQ.2) GO TO 120
      CAY = 3.*GUESS(L-1) - 2.*GUESS(L-2)
      NN = NN + 1
      IF (NN.GT.10) GO TO 100
      GO TO 20
130   RL = L
      RM = M
      CAY = CAY*(RL/RM)
      JJ = JJ + 1
      IF (JJ.GT.10) GO TO 100
      GO TO 20
120   CAY = 1.5*GUESS(L)
      GO TO 20
100   WRITE (6,160)
160   FORMAT (10X,'PROGRAM UNABLE TO DETERMINE CORRECT MODE. WILL PROCEE
     *D TO NEXT FREQUENCY.')
      GO TO 10
309   WRITE(6,310)
310   FORMAT (10X,'VALUE OF W IS TOO HIGH. FURTHER MODES WILL BE ABORTED
     *. PROGRAM WILL PROCEED TO NEXT FREQUENCY.',/)
```

```
10 CONTINUE
   RETURN
   END
```

LIST OF REFERENCES

Fjeldstad, J.E., "Interne Wellen." Geophys. Publ., Vol. 10, No. 6, pp. 1-35 (1933).

LaFond, E.C., "Internal Waves, Part I." The Sea (Ed. M.N. Hill), Volume I, pp. 731-751 (1962).

Oceanographic Cruise Report of Cruise NPS 1201-71, DeSteiguer (T-AGOR 12), August 10-20, 1970.

Wunsch, C., and Dahlen, J., "Preliminary Results of Internal Wave Measurements in the Main Thermocline at Bermuda." Journal of Geophysical Research, Vol. 75, No. 30, pp. 5899-5908 (1970).

Zalkan, R.L., Observation of High-Frequency Internal Waves in the Pacific Ocean, Ph.D. Thesis, University of California, San Diego, (1968).

INITIAL DISTRIBUTION LIST

		No. Copies
1.	Defense Documentation Center Cameron Station Alexandria, Virginia 22314	2
2.	Library, Code 0212 Naval Postgraduate School Monterey, California 93940	2
3.	Department of Oceanography, Code 58 Naval Postgraduate School Monterey, California 93940	3
4.	Asst. Professor J.A. Galt Department of Oceanography, Code 58 Naval Postgraduate School Monterey, California 93940	3
5.	Assoc. Professor J.J. von Schwind Department of Oceanography, Code 58 Naval Postgraduate School Monterey, California 93940	1
6.	Oceanographer of the Navy The Madison Building 732 N. Washington Street Alexandria, Virginia 22314	1
7.	Naval Oceanographic Office Attention: Library Washington, D.C. 20390	1
8.	Dr. Ned Ostenso Office of Naval Research Code 480D Arlington, Virginia 22217	1
9.	LCdr. W.J. Lounsbery KWESTEVDET Key West, Florida 33040	2
10.	Lt. Jose Saldanha Instituto Hidrografico Ministerio da Marinha Lisboa, Portugal	1

Security Classification

DOCUMENT CONTROL DATA - R & D

(Security classification of title, body of abstract and indexing annotation must be entered when the overall report is classified)

1. ORIGINATING ACTIVITY *(Corporate author)* Naval Postgraduate School Monterey, California 93940	2a. REPORT SECURITY CLASSIFICATION Unclassified 2b. GROUP

3. REPORT TITLE

Response of a Water Column to Internal Waves of Known Frequencies

4. DESCRIPTIVE NOTES *(Type of report and, inclusive dates)*
Master's Thesis; September 1971

5. AUTHOR(S) *(First name, middle initial, last name)*

William J. Lounsbery

6. REPORT DATE	7a. TOTAL NO. OF PAGES	7b. NO. OF REFS
September 1971	60	5

8a. CONTRACT OR GRANT NO. b. PROJECT NO. c. d.	9a. ORIGINATOR'S REPORT NUMBER(S) 9b. OTHER REPORT NO(S) *(Any other numbers that may be assigned this report)*

10. DISTRIBUTION STATEMENT

Approved for public release; distribution unlimited.

11. SUPPLEMENTARY NOTES	12. SPONSORING MILITARY ACTIVITY
	Naval Postgraduate School Monterey, California 93940

13. ABSTRACT

A program is developed to ascertain the response of a frictionless water column to internal waves. Using density strata as input, the program selects ten frequencies at equal intervals within a spectrum of internal waves bounded above by the maximum Vaisala frequency and below by the inertial frequency. Ray paths within the column are plotted for these frequencies. The first ten normal modes for each frequency are computed. The first three modes are plotted.

DD FORM 1 NOV 65 1473 (PAGE 1)

S/N 0101-807-6811

Security Classification

A-31408

KEY WORDS	LINK A		LINK B		LINK C	
	ROLE	WT	ROLE	WT	ROLE	WT
Internal waves						
Ray paths						
normal modes						
computer programs						
numerical methods						

Thesis
L827
c.2

131205

Lounsbery
Response of a water column to internal waves of known frequencies.